4.- DERIVABILIDAD DE UNA FUNCION REAL DE VARIABLE REAL APLICACIONES PRÁCTICAS

EJERCICIO 1

Partiendo de la función f(x) = $\sqrt{x+3}$, hallar aproximadamente el valor de $\sqrt{3,98}$ utilizando el concepto de diferencial

Definimos la función a partir de la cual debemos calcular el valor aproximado

```
f[x_] = √(x + 3);
```

Definimos la función df aplicando el concepto df = f '(x) Δx

```
df[x_, Δx_] = f'[x] * Δx
```

$$\frac{\Delta x}{2\sqrt{3+x}}$$

Definimos la función valor aproximado aplicando el concepto f (x+Δx) - f (x) ≃ f '(x) Δx

```
valaproxf[x_, Δx_] = df[x, Δx] + f[x]
```

$$\sqrt{3+x}+\frac{\Delta x}{2\sqrt{3+x}}$$

Hacemos en la función valor aproximado las asignaciones x = 1 y Δx = - 0.02

```
valaproxf[1, -0.02]
```

1.995

Comprobamos que el valor real es casi igual al aproximado

```
f[1 - 0.02]
```

1.99499

EJERCICIO 2

Calcular la recta tangente a la curva y = $\frac{2x}{x^2+1}$ en el punto x = $\frac{1}{2}$, dibujando la función y su recta tangente en ese punto.

Definimos la función f(x) y calculamos su primera derivada

$$f[x\_] = \frac{2\,x}{x^2+1};$$

Definimos la función tangente utilizando la definición de recta tangente en un punto

```
tan[x_, a_] = f'[a] * (x - a) + f[a]
```

$$\frac{2\,a}{1+a^2} + \left(-\frac{4\,a^2}{\left(1+a^2\right)^2} + \frac{2}{1+a^2}\right)(-a+x)$$

Calculamos la recta tangente a la función f/x) que pasa por el punto $x = \frac{1}{2}$

$$\tan\left[x, \frac{1}{2}\right]$$

$$\frac{4}{5} + \frac{24}{25}\left(-\frac{1}{2} + x\right)$$

Dibujamos la función f(x) y su recta tangente en el punto $x = \frac{1}{2}$, comprobando que efectivamente es tangente en ese punto

```
Plot[{f[x], tan[x, 1/2]}, {x, -1, 2}, PlotStyle →
  {{Orange, Thickness[0.01]}, {Purple, Thickness[0.01]}}]
```



EJERCICIO 3

Dada la función y = f(x) expresada en forma implícita por la ecuación $2y = 1+xy^3$, calcular y´e y´´ en el punto P=(1,1)

Definimos la función f(x) y calculamos la derivada implícita de y respecto de x

```
ecu = 2 y == 1 + x * y^3;
Dt[ecu, x]
```

$2\,\mathrm{Dt}[y, x] == y^3 + 3\,x\,y^2\,\mathrm{Dt}[y, x]$

Despejamos y' en la ecuación y obtenemos su valor en el punto P=(1,1)

```
s = Solve[%, Dt[y, x]]
```

$\left\{\left\{\mathrm{Dt}[y, x] \to -\frac{y^3}{-2 + 3\,x\,y^2}\right\}\right\}$

```
yprima = Dt[y, x] /. Flatten[s]
```

$-\frac{y^3}{-2 + 3\,x\,y^2}$

```
yprima = yprima /. {x -> 1, y -> 1}
```

-1

Calculamos la segunda derivada implícita de y respecto de x

```
Dt[ecu, {x, 2}]
```

$2\,\mathrm{Dt}[y, \{x, 2\}] == 6\,y^2\,\mathrm{Dt}[y, x] + x\left(6\,y\,\mathrm{Dt}[y, x]^2 + 3\,y^2\,\mathrm{Dt}[y, \{x, 2\}]\right)$

Despejamos y'' en la ecuación y obtenemos su valor en el punto P=(1,1) sustituyendo el valor de y' obtenido anteriormente

```
s = Solve[%, Dt[y, {x, 2}]]
```

$\left\{\left\{\mathrm{Dt}[y, \{x, 2\}] \to -\frac{6\left(y^2\,\mathrm{Dt}[y, x] + x\,y\,\mathrm{Dt}[y, x]^2\right)}{-2 + 3\,x\,y^2}\right\}\right\}$

```
s = s /. Dt[y, x] → yprima
```

$\left\{\left\{Dt[y, \{x, 2\}] \to -\frac{6\,(x\,y - y^2)}{-2 + 3\,x\,y^2}\right\}\right\}$

```
ysegunda = Dt[y, {x, 2}] /. Flatten[s]
```

$-\frac{6\,(x\,y - y^2)}{-2 + 3\,x\,y^2}$

```
ysegunda = ysegunda /. {x -> 1, y -> 1}
```

0

EJERCICIO 4

Calcular la ecuación de las rectas tangentes a la curva $2x^2 - 2xy + y^2 + x + 2y + 2 = 0$ en los puntos de abscisa x = -1

Definimos la curva y en ella sustituimos el valor x=-1 para obtener los puntos de la misma por los que han de pasar las rectas tangentes

```
cur = 2 x^2 - 2 x * y + y^2 + x + 2 y + 2 == 0;
x0 = -1;
cur /. x → x0
```

$3 + 4\,y + y^2 == 0$

```
s = Solve[%, y]
```

{{y → -3}, {y → -1}}

```
y1 = y /. s[[1]]
```

-3

```
y2 = y /. s[[2]]
```

-1

Derivamos implícitamente la función para obtener la pendiente de la recta tangente

```
Dt[cur, x]
```

$1 + 4x - 2y + 2\,Dt[y, x] - 2x\,Dt[y, x] + 2y\,Dt[y, x] == 0$

```
s = Solve[%, Dt[y, x]]
```

$\left\{\left\{Dt[y, x] \to \frac{1 + 4x - 2y}{2(-1 + x - y)}\right\}\right\}$

```
pendiente[x_, y_] = Dt[y, x] /. Flatten[s]
```

$\frac{1 + 4x - 2y}{2(-1 + x - y)}$

Sustituimos en la ecuación de la recta tangente la pendiente para cada uno de los puntos calculados

```
rectatangente1 = y == pendiente[x0, y1] (x - x0) + y1 // Simplify
```

$3x == 3 + 2y$

```
rectatangente2 = y == pendiente[x0, y2] (x - x0) + y2 // Simplify
```

$x == 1 + 2y$

EJERCICIO 5

Indicar si la función f(x) = $\begin{cases} x^2 \operatorname{sen} \frac{1}{x} & \text{si} \quad x \neq 0 \\ 0 & \text{si} \quad x = 0 \end{cases}$ es derivable en x=0

Definimos la función y vemos gráficamente como se comporta en el punto x=0

```
f[x_] = If[x ≠ 0, x^2 * Sin[1/x], 0];
```

```
Plot[f[x], {x, -Pi / 48, Pi / 48}]
```



Calculamos las derivadas laterales en el punto x=0

$$\text{Limit}\left[\frac{f[0+h]-f[0]}{h},\ h\to 0,\ \text{Direction}\to -1\right]$$

0

$$\text{Limit}\left[\frac{f[0+h]-f[0]}{h},\ h\to 0,\ \text{Direction}\to 1\right]$$

0

Como las derivadas laterales en x=0 son iguales, existe la derivada en x = 0 y por tanto la función es derivable en dicho punto

EJERCICIO 6

Dada la función f(x) = $\begin{cases} \frac{ax+b}{x^2+2x+1} & \text{si} \quad x > 0 \\ \frac{x^2+1}{x-1} & \text{si} \quad x \le 0 \end{cases}$ **, obtener el valor de los parámetros reales a y b para que f(x) sea continua y derivable en el punto x=0**

Definimos la función f(x)

```
f[x_] = If[x > 0, (a * x + b) / (x^2 + 2 * x + 1), (x^2 + 1) / (x - 1)];
```

Calculamos el valor de b para que exista el límite de la función en x=0

```
limitederecha = Limit[f[x], x -> 0, Direction → -1]
```

b

```
limiteizquierda = Limit[f[x], x -> 0, Direction -> 1]
```

-1

```
Solve[limiteizquierda == limitederecha]
```

{{b → -1}}

```
b = %[[1, 1, 2]]
```

-1

Calculamos el valor de a para que la función sea derivable en x=0

derivadaderecha = Limit$\left[\frac{f[0+h]-f[0]}{h}, h \to 0, \text{Direction} \to -1\right]$

2 + a

derivadaizquierda = Limit$\left[\frac{f[0+h]-f[0]}{h}, h \to 0, \text{Direction} \to 1\right]$

-1

```
Solve[derivadaderecha == derivadaizquierda]
```

{{a → -3}}

```
a = %[[1, 1, 2]]
```

-3

EJERCICIO 7

Dada la función f (x) = Sen (x) Ln(x) utiliza su diferencial para calcular el valor aproximado de f (1,05)

Definimos la función a partir de la cual debemos calcular el valor aproximado

```
f[x_] = Sin[x] Log[x];
```

Utilizamos el procedimiento visto en el ejercicio 1

```
df[x_, Δx_] = f'[x] * Δx
```

$$\Delta x \left(\mathrm{Cos}[x]\,\mathrm{Log}[x] + \frac{\mathrm{Sin}[x]}{x}\right)$$

```
valaproxf[x_, Δx_] = df[x, Δx] + f[x]
```

$$\mathrm{Log}[x]\,\mathrm{Sin}[x] + \Delta x \left(\mathrm{Cos}[x]\,\mathrm{Log}[x] + \frac{\mathrm{Sin}[x]}{x}\right)$$

Hacemos en la función valor aproximado las asignaciones $x = 1$ y $\Delta x = 0.05$

```
valaproxf[1, 0.05]
```

0.0420735

Comprobación

```
f[1 + 0.05]
```

0.0423217